
Tijdmenu is een programma dat speciaal gemaakt is voor de NMS 8280 van Philips. Hoewel het 
programma ook werkt op alle andere MSX2-computers, komt het alleen met die speciale 
machine echt tot zijn recht. 

Wat het doet is simpel: een soort tijd- en datumstempel op het scherm zetten. Echt functioneel 
wordt Tijdmenu pas, als het gebruikt kan worden in combinatie met een 'superimposed' beeld. 
Dat wil zeggen: een beeld met een video-achtergrond en een   computer-voorgrond. Die door de 
computer gemaakte voorgrond is nu de tijdcode, die naar believen ingesteld kan worden. 

Zo'n tijdcode is onontbeerlijk voor video-hobbyisten en professionals. We kunnen ons 
herinneren dat we een paar jaar terug — toen de redacties van MSX/MS-DOS Computer 
Magazine en Video Uit en Thuis nog in één pand gehuisvest waren — we ons vaak genoeg aan 
die tijdcodes hebben geërgerd. 

Op hun redactie stond namelijk een flinke kast, die altijd gevuld was met de nieuwste video-
films. Zogenaamd ter recensie, maar iedereen — ook wij — putte met regelmaat uit die 
voorraad. Meer dan de helft van die cassettes was echter voorzien van die infernale tijd code! 
Leuk, een film met een per tiende seconde bijgehouden tijd stempel, dat meestal pontificaal in 
beeld stond. En nog zonder ondertitels bovendien ook! 

Werkkopie 

Waar we dan naar keken, dat waren dan ook werkkopieën. Films in een vorm die nooit zo bij het 
publiek zal komen.Immers, recensenten krijgen de film altijd eerder, voordat de ondertiteling 
klaar is. Voor dat ondertitelen bijvoorbeeld is zo'n tijdcode onontbeerlijk. Men kan precies 
aflezen op welk moment een bepaalde regel moet verschijnen, en daar later de 
ondertitelingsmachine mee programmeren. 

Ook voor montagewerk is zo'n kopie met tijdcode ideaal; zo kunnen scènes precies getimed 
worden. Vandaar dat we voor de liefhebbers Tijdmenu hebben geschreven, een simpel 
programma dat echter voor de video hobbyist werkelijk ideaal gereedschap vormt. 

NMS 8280 

De NMS 8280 is al eerder in dit blad beschreven vanwege zijn opmerkelijke video-kwaliteiten. 
Om het geheugen van de lezer op te frissen: de NMS 8280 van Philips is een MSX 
'beeldverwerker' met twee video uitgangen en twee video-ingangen. Deze ingangen worden 
synchroon gekoppeld aan het videobeeld van de computer zelf. 

Daardoor is het mogelijk om een computersignaal — bijvoorbeeld titelteksten, tijd codes of 
digitalisaties — aan het oorspronkelijke videobeeld, dat via de externe video ingang de computer 
inkomt, toe te voegen. De NMS 8280 kent, zoals iedere MSX2 machine, de nodige kleuren. 



Alleen kleurcode 0, die normaal gezien als zwart op het scherm verschijnt, daar is iets bijzonders 
mee. Kleur 0 is namelijk helemaal geen echte kleur, maar de code waardoor de MSX weet dat 
dat schermdeel als 'transparant gezien moet worden. Dat heeft pas effect als er ook een extern 
video beeld wordt aangeboden en we bovendien in Basic het commando SET VIDEO 2 gegeven 
hebben. Dan zal op dit als transparant gekenmerkte gedeelte het externe videobeeld zichtbaar 
zijn. Dit noemt men in goed Nederlands een 'superimposed' beeld. 

Toepassingen 

Van deze laatstgenoemde superimpose instelling wordt in het programma Tijdmenu effectief 
gebruik gemaakt om een tijdcode over het video-beeld heen op het beeldscherm te toveren. De 
computer is nu een stukje gereedschap met diverse handige foefjes voor video-montage of video 
timing geworden. Natuurlijk kan men met een tweede recorder vervolgens het uitgangssignaal 
van de NMS 8280 weer opnemen, om zo een echte werkkopie aan te maken 

Maar ook als we alleen wat shots willen monteren is Tijdmenu een ideaal middel om de tijden te 
meten. 

Al met al kent Tijdmenu vier mogelijkheden: 

- De nog te monteren video opnamen van een tijdcode voorzien; 

- Hulpje bij het maken van opnametabel (oftewel shotlist); 

- Het direct opnemen van een tijdcode bij thuis- (of studio-) opnamen en 

- Tijdmeting bij thuis opnamen via de monitor. 

De vele manieren van aansluiten is de NMS 8280 'eigen'. Raadpleeg voor de juiste configuratie 
en aansluitingen de handleiding van de NMS 8280. Een setje verloop kabels kan wel 
noodzakelijk blijken! 

Instellingen 

Als het programma in de NMS 8280 -of andere MSX2-computer — is geladen, gaan we 
welgemoed aan de slag. Eerst verschijnt het menu van het programma, waarin alle instellingen 
overzichtelijk gerangschikt zijn. Zorg dat de recorders correct zijn aangesloten en druk de toets 
's' in. Vervolgens verschijnt de tijdcode op het beeldscherm; met de cursortoets dirigeert men de 
tijdcode naar de gewenste plaats op het beeld. Door middel van de cursortoetsen en enkele 
andere letter- en functietoetsen is Tijdmenu te besturen. 

Het activeren van een functie in het menu van het programma is simpel: de rode balk met de 
cursor toetsen op de gewenste functie zetten en de RETURN- of ENTER-toets indrukken. 
Voor het uitzetten van diezelfde functie drukt men nogmaals op de RETURN toets. De video 
opnamen zonder tijd code of datum-aanduiding kunnen eenvoudig gekopieerd worden met de 
huidige computertijd, de vast ingestelde tijd van de computer die altijd gelijk blijft lopen, of een 
zelfgekozen tijd. De ingestelde tijd van de computer verandert overigens niet als men de tijd in 
het programma aanpast. 

De kleuren kunnen ingesteld worden met de cursor toetsen, evenals de plaats waar de tijdcode op 
het beeldscherm moet staan. De kleur van de tijdcode is bij het opstarten van het programma fel 
geel met een blauwe schaduw kleur. Deze schaduw kleur moet voorkomen dat de tijdcode 
wegvalt tegen de kleuren van de video opnamen. Er zijn zo'n vijftien kleuren beschikbaar. In 



plaats van een schaduw kleur kan men ook een kleurvlak achter de tijdcode plaatsen. Dit is 
echter alleen mogelijk bij de T(ijd) en D(atum)' instelling en niet als men kiest voor de 
mogelijkheid T(ijd) of D(atum)\ Bij de transparant-instelling (oftewel superimpose) is het 
natuurlijk niet mogelijk een andere achtergrondkleur dan zwart te benoemen. Als er een andere 
kleur aanstaat wordt deze automatisch in zwart veranderd. 



Menu functies 

 

RETURN Activeert geselecteerde menu functie 

Cursortoetsen Menubalk verplaatsen Tijd instellen Kleuren instellen 

S- of s-toets Start tijdcode 

Start Brengt de tijdcode op het beeldscherm 

Seconden Seconden instellen met cursortoetsen 

Minuten Minuten instellen met cursortoetsen 

Uren Uren instellen met cursortoetsen 

Reset tijd De tijdcode op nul stellen 

T+D Tijdcode bestaat uit huidige tijd en de huidige datum 

T Alleen huidige tijd 

D Alleen huidige datum 

Cijferkleur Voorgrond kleur en schaduw kleur van de tijdcode  

met cursor toetsen  

Achtergrond vlak Als T+D gekozen zijn dan vervalt de schaduwkleur en is de 
achtergrond kleur — een effen rechthoekige achtergrond — 
in te stellen met de cursortoetsen 

Schermkleur Als de achtergrond kleur niet transparant is, kan de 
achtergrond kleur ingesteld worden met de cursor toetsen 

Transparant Superimpose aan/uit 

Snel/Langzaam Instelling van de aanwijsbalk snel en langzaam knipperend; 
de verplaatsing van de tijdcode is per l pixel of per 16 pixels 
instelbaar. Bovendien is in de langzame stand de huidige tijd 
op het menubeeld zichtbaar. 

Schermfuncties 

 

ESC Terug naar menu als de tijdcode op het beeldscherm staat 

Cursortoetsen  Verplaatsen in vier richtingen van de tijdcode over het beeldscherm 

Spatiebalk Zet de waarden van de tijdcode op nul 

 



Tijdmenu.bas 

10 REM Tijdmenu 
20 REM 
30 REM MSX/DOS Computer Magazine 
40 REM 
50 DEFINT A-Z: DIM B(16) 
60 K=253: R=0: IS=0: AS=0: DT=1: KT=3: SL=0: TT=0 
70 SCREEN 8: SET VIDEO 0: OPEN "grp:" FOR OUTPUT AS#1 
80 ' menu afdrukken*********************************************************' 
90 COLOR 255,74,74: CLS 
100 FOR TE=1 TO 17 
110    PSET(15,0+TE*12),74: READ A$: COLOR 255,37: PRINT#1,A$;SPC(21-LEN(A$)) 
120 NEXT TE 
130 IF SL=0 THEN TE=1 ELSE TE=17 
140 IF IS=0 THEN GET TIME T$: U$=LEFT$(T$,2): M$=MID$(T$,4,2): M=VAL(M$): 
U=VAL(U$): S=VAL(RIGHT$(T$,2)) 
150 LINE (140,0)-(142,211),74,BF: COLOR 255,37 
160 PSET (150,36),37: PRINT#1,S 
170 PSET (150,48),37: PRINT#1,M 
180 PSET (150,60),37: PRINT#1,U 
190 PSET (151,108),37: IF DT=1 THEN PRINT#1,"T+D" 
200 PSET (150,108),37: IF DT=0 THEN PRINT#1,SPC(1);"T";SPC(1): PSET 
(15,144),37 
210 PSET (150,108),37: IF DT=2 THEN PRINT#1,SPC(1);"D";SPC(1): PSET 
(15,156),37 
220 PSET (158,132),K: COLOR K,255-K: PRINT#1,CHR$(200) 
230 PSET (158,144),R: COLOR R,37: PRINT#1,CHR$(200) 
240 PSET (158,156),37: COLOR AK,37: PRINT#1,CHR$(200) ELSE PRINT#1,SPC(1) 
250 PSET(155,180),37: COLOR 255,37 
260 IF TT=1 THEN PRINT#1,"Ja";SPC(1) ELSE PRINT#1,"Nee" 
270 ' menubalk + keuzen****************************************************** 
280 Z=STICK(0): IF Z=1 THEN TE=TE-1: IF TE<1 THEN TE=1 
290 IF Z=5 THEN TE=TE+1: IF TE>17 THEN TE=17 
300 H$=INKEY$ 
310 IF H$="s" OR H$="S" THEN GOTO 360 
320 LINE (15,TE*12)-(21*8+14,7+TE*12),28,BF,XOR: FOR T=0 TO 25: NEXT T 
330 LINE (15,TE*12)-(21*8+14,7+TE*12),28,BF,XOR 
340 IF H$=CHR$(13) THEN ON TE GOTO 
360,270,670,760,850,270,930,270,950,270,990,1130,1270,270,1400,270,1450 
350 IF SL=1 THEN A=1: GOTO 140 ELSE A=16: GOTO 280 
360 ' tijd routine*********************************************************** 
370 COLOR 255,AK,AK: CLS 
380 GET TIME T$: GET DATE D$: I$=RIGHT$(T$,2): T=VAL(I$) 
390 SET PAGE 0,1: COLOR K,R: CLS 
400 IF IS=1 AND (DT=1 OR DT=0) THEN GOTO 410 ELSE GOTO 420 
410 PSET (0,0),0: PRINT#1,USING "##:##:##";U,M,S: GOTO 430 
420 IF DT<2 THEN PSET (0,0),0: PRINT#1,T$ 
430 IF DT=1 OR DT=2 THEN GOTO 440 ELSE GOTO 450 
440 PSET (0,10),0: PRINT#1,MID$(D$,4,2);"/";LEFT$(D$,2);RIGHT$(D$,3) 
450 IF T<>B THEN S=S+1: B=T 
460 IF M>59 THEN M=0: U=U+1 
470 IF S>59 THEN S=0: M=M+1 
480 IF R>0 THEN GOTO 570 
490 COPY (0,0)-(80,18),1 TO (101,101),1,TPRESET 
500 COPY (0,0)-(80,18),1 TO (101,100),1,TPRESET 
510 COPY (0,0)-(80,18),1 TO (101,99),1,TPRESET 
520 COPY (0,0)-(80,18),1 TO (99,101),1,TPRESET 
530 COPY (0,0)-(80,18),1 TO (99,100),1,TPRESET 
540 COPY (0,0)-(80,18),1 TO (99,99),1,TPRESET 



550 COPY (0,0)-(80,18),1 TO (100,99),1,TPRESET 
560 COPY (0,0)-(80,18),1 TO (100,101),1,TPRESET 
570 COPY (0,0)-(80,18),1 TO (100,100),1,TPSET 
580 SET PAGE 0,0 
590 COPY (99,99)-(161,117),1 TO (X,Y),0 
600 H$=INKEY$ 
610 IF H$=CHR$(32) THEN IS=1:AS=1: S=0: M=0: U=0 
620 IF H$=CHR$(27) THEN RESTORE: GOTO 90 
630 D=STICK(0): IF D=0 THEN GOTO 380 
640 LINE (X,Y)-(X+62,Y+18),AK,BF: GOSUB 1490 
650 GOTO 380 
660 ' seconden instellen***************************************************** 
670 XX=X: YY=Y: Y=S: IS=1: 'seconden instellen 
680 COLOR 28,37: PSET (150,TE*12),37: PRINT#1,S 
690 S=Y: A=-1 
700 IF S<0 THEN S=59: Y=59 
710 IF S>59 THEN S=0: Y=0 
720 IF INKEY$=CHR$(13) THEN X=XX: Y=YY: GOTO 140 
730 GOSUB 1490 
740 GOTO 680 
750 ' minuten instellen****************************************************** 
760 XX=X: YY=Y: Y=M: IS=1 
770 COLOR 28,37: PSET (150,TE*12),37: PRINT#1,M 
780 M=Y: A=-1 
790 IF M<0 THEN M=59: Y=59 
800 IF M>59 THEN M=0: Y=0 
810 IF INKEY$=CHR$(13) THEN X=XX: Y=YY: GOTO 140 
820 GOSUB 1490 
830 GOTO 770 
840 ' uren instellen********************************************************* 
850 XX=X: YY=Y: Y=U: IS=1 
860 COLOR 28,37: PSET (150,TE*12),37: PRINT#1,U 
870 U=Y: A=-1 
880 IF U<0 THEN U=23: Y=23 
890 IF U>23 THEN U=0: Y=0 
900 IF INKEY$=CHR$(13) THEN X=XX: Y=YY: GOTO 140 
910 GOSUB 1490 
920 GOTO 860 
930 ' reset ingestelde tijd************************************************** 
940 IS=1: AS=1: S=0: M=0: U=0: RESTORE: GOTO 140 
950 ' tijd en datum 
********************************************************** 
960 IS=0: DT=DT+1: IF DT>2 THEN DT=0 
970 IF DT=0 OR DT=2 THEN R=0: AK=0: GOTO 140 
980 GOTO 140 
990 ' cijferkleur instellen************************************************** 
1000 GOSUB 1600 
1010 XX=X: YY=Y 
1020 PSET (158,TE*12),K: COLOR K,255-K: PRINT#1,CHR$(200) 
1030 LINE (15,TE*12)-(139,TE*12+7),28,BF,XOR: FOR T=0 TO 50: NEXT T 
1040 LINE (15,TE*12)-(139,TE*12+7),28,BF,XOR 
1050 RR=Y: A=-1 
1060 IF RR>15 THEN RR=15: Y=15 
1070 IF RR<1 THEN RR=1: Y=1 
1080 K=B(RR) 
1090 IF INKEY$=CHR$(13) THEN X=XX: Y=YY: GOTO 290 
1100 GOSUB 1490 
1110 GOTO 1020  
1120 ' achtergrondvlak kleur instellen*************************************** 
1130 XX=X: YY=Y 
1140 GOSUB 1600 
1150 IF DT<>1 THEN R=0: GOTO 140 



1160 LINE (15,TE*12)-(139,TE*12+7),28,BF,XOR: FOR T=0 TO 50: NEXT T 
1170 LINE (15,TE*12)-(139,TE*12+7),28,BF,XOR 
1180 PSET (158,TE*12),R: COLOR R,37: PRINT#1,CHR$(200) 
1190 RR=Y: A=-1 
1200 IF RR>14 THEN RR=15: Y=15 
1210 IF RR<1 THEN RR=0: Y=0 
1220 R=B(RR) 
1230 IF INKEY$=CHR$(13) THEN X=XX: Y=YY: GOTO 140 
1240 GOSUB 1490 
1250 GOTO 1150 
1260 '  achtergrond schermkleur instellen *********************************** 
1270 XX=X: YY=Y 
1280 GOSUB 1600 
1290 IF DT<>1 OR TT=1 THEN AK=0: GOTO 140 
1300 LINE (15,TE*12)-(139,TE*12+7),28,BF,XOR: FOR T=0 TO 50: NEXT T 
1310 LINE (15,TE*12)-(139,TE*12+7),28,BF,XOR 
1320 PSET (158,TE*12),AK: COLOR AK,37: PRINT#1,CHR$(200) 
1330 RR=Y: A=-1 
1340 IF RR>15 THEN RR=15: Y=15 
1350 IF RR<1 THEN RR=0: Y=0 
1360 AK=B(RR) 
1370 IF INKEY$=CHR$(13) THEN X=XX: Y=YY: GOTO 140 
1380 GOSUB 1490: GOTO 1290 
1390 GOTO 1290 
1400 'transparant 
1410 IF TT=0 THEN GOTO 1420 ELSE GOTO 1440 
1420 AK=0 
1430 SET VIDEO 2: TT=1: GOTO 140 
1440 SET VIDEO 0: TT=0: GOTO 140 
1450 ' snel/langzaam********************************************************* 
1460 IF SL=1 THEN SL=0 ELSE SL=1: GOTO 140 
1470 GOTO 280 
1480 ' cursor routine ******************************************************* 
1490 D=STICK(0) 
1500 IF D=1 THEN Y=Y-1*A: IF Y<0 THEN Y=0 
1510 IF D=3 THEN X=X+1*A: IF X>192 THEN X=192 
1520 IF D=5 THEN Y=Y+1*A: IF Y>192 THEN Y=192 
1530 IF D=7 THEN X=X-1*A: IF X<1 THEN X=0 
1540 RETURN 
1550 ' menu data 
1560 DATA START," ","Seconden","Minuten","Uren"," ","Reset TIJD"," " 
1570 DATA "DATUM en TIJD"," ","Cijferkleur","Achtergrond Vlak" 
1580 DATA "Schermkleur"," ","Transparant"," ","Snel/Langzaam" 
1590 ' kleurnumering 
1600 RESTORE 1600 
1610 FOR AA=0 TO 15 
1620    READ KK: B(AA)=KK 
1630 NEXT AA 
1640 RETURN 
1650 DATA 0,1,2,3,5,10,31,8,16,28,36,72,224,252,253,254 


	Tijdmenu: video-programma
	
	
	
	
	
	Scanned, ocr’ed and converted to PDF by HansO, 20





	Werkkopie
	NMS 8280
	Toepassingen
	Instellingen
	Menu functies
	Schermfuncties
	Scanned, ocr’ed and error checked with checksumpr



